a strong form of reflexivity

Johan G. F. Belinfante
2008 February 12

In[1]:= **SetDirectory["l:"]; << goedel.08feb10a; << tools.m**

```
:Package Title: goedel.08feb10a                    2008 February 10 at 2:15 p.m.

It is now:  2008 Feb 12 at 16:40

Loading Simplification Rules

TOOLS.M                                            Revised 2008 January 22

weightlimit = 40
```

summary

This notebook is concerned with a careful study of the concept of a reflexive relation. Patrick Suppes introduced the following definition for a strong form of reflexivity of a relation **x** on a set **y**. (Actually, both **x** and **y** can be arbitrary classes here.)

In[2]:= **assert[forall[t, implies[member[t, y], member[pair[t, t], x]]]]**

Out[2]= subclass[y, fix[x]]

Reference: (see Definition 10 on page 69).

In[3]:= **"Patrick Suppes, *Axiomatic Set Theory*, Dover Publications, New York, 1972.";**

We shall say that **x** is **strongly reflexive** in **y** if the condition **subclass[y, fix[x]]** holds. In the **GOEDEL** program, the (unary) predicate **REFLEXIVE[x]** is defined by the condition

In[4]:= **subclass[x, cart[fix[x], fix[x]]]**

Out[4]= REFLEXIVE[x]

In this notebook it is shown that if **x** is strongly reflexive in **y**, then the symmetric restriction **intersection[x, cart[y,y]]** is reflexive in the sense of the **GOEDEL** program, but not conversely. The two conditions do agree for the special case that **y** is the union of the domain and range of **x**. This is usually called the **field** of **x**, but to avoid confusion with other meaning of the term field, the following acronym will be introduced for this union:

In[5]:= **udora[x_] := union[domain[x], range[x]]**

The restriction **restrict[x, y, z] = intersection[x, cart[y, z]]** of a relation is rewritten by the **GOEDEL** program as a composite:

```
In[6]:= intersection[x, cart[y, z]]

Out[6]= composite[id[z], x, id[y]]
```

In particular, the symmetric restriction of a class **x** to the union of its domain and range is the relational part of **x**, that is, the class of all ordered pairs that belong to **x**.

```
In[7]:= restrict[x, udora[x], udora[x]]

Out[7]= composite[Id, x]
```

strong reflexivity => reflexivity of the restriction

Lemma.

```
In[8]:= SubstTest[implies, equal[y, intersection[t, fix[x]]],
          or[equal[0, intersection[y, image[x, intersection[y, complement[fix[x]]]]]],
           not[subclass[y, fix[x]]]], t → y] // Reverse

Out[8]= or[equal[0, intersection[y, image[x, intersection[y, complement[fix[x]]]]]],
          not[subclass[y, fix[x]]]] == True

In[9]:= (% /. {x → x_, y → y_}) /. Equal → SetDelayed
```

Theorem. Strong reflexivity implies reflexivity of the symmetric restriction.

```
In[10]:= or[not[subclass[y, fix[x]]], REFLEXIVE[composite[id[y], x, id[y]]]] // AssertTest

Out[10]= or[not[subclass[y, fix[x]]], REFLEXIVE[composite[id[y], x, id[y]]]] == True

In[11]:= or[not[subclass[y_, fix[x_]]], REFLEXIVE[composite[id[y_], x_, id[y_]]]] := True
```

Counterexample. The converse statement does not hold. The strong assertion **subclass[y, fix[x]]** does not follow from the weaker statement **REFLEXIVE[restrict[x,y,y]]**.

```
In[12]:= (implies[REFLEXIVE[composite[id[y], x, id[y]]], subclass[y, fix[x]]]) /.
          {x → EQUIDIFF, y → V}

Out[12]= False
```

Comment. The statement that **x** is strongly reflexive on **udora[x]** is equivalent to the statement that the symmetric restriction of **x** to **udora[x]** is reflexive.

```
In[13]:= subclass[udora[x], fix[x]]

Out[13]= REFLEXIVE[composite[Id, x]]
```

the class of sets for which the restriction of x is reflexive

In this section some properties of the class of sets on which the symmetric restriction of **x** is reflexive are derived. The following formula for this class is readily derived: **fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]**. Indeed, one has:

```
In[14]:= member[y, fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]]

Out[14]= and[member[y, V], REFLEXIVE[composite[id[y], x, id[y]]]]
```

Lemma. Simplification rule.

```
In[15]:= subclass[intersection[y, image[inverse[x], w]], z] // AssertTest // Reverse

Out[15]= equal[0, intersection[w, image[x, intersection[y, complement[z]]]]] ==
          subclass[intersection[y, image[inverse[x], w]], z]

In[16]:= equal[0, intersection[w_, image[x_, intersection[y_, complement[z_]]]]] :=
          subclass[intersection[y, image[inverse[x], w]], z]
```

Theorem. The statement that the restriction of **x** to **y** is symmetric can be rewritten as follows:

```
In[17]:= SubstTest[subclass, t, cartsq[fix[t]], t → restrict[x, y, y]] // Reverse

Out[17]= and[subclass[intersection[y, image[x, y]], fix[x]],
           subclass[intersection[y, image[inverse[x], y]], fix[x]]] ==
          REFLEXIVE[composite[id[y], x, id[y]]]

In[18]:= and[subclass[intersection[y_, image[x_, y_]], fix[x_]],
           subclass[intersection[y_, image[inverse[x], y_]], fix[x_]]] :=
          REFLEXIVE[composite[id[y], x, id[y]]]
```

The class of sets **y** for which each of these inclusions holds is a class of cliques. This explains the following normalization rule:

```
In[19]:= fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]] // Normality // Reverse

Out[19]= intersection[cliques[union[cart[fix[x], V], complement[x]]],
           cliques[union[cart[fix[x], V], complement[inverse[x]]]]] ==
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]

In[20]:= intersection[cliques[union[cart[fix[x_], V], complement[x_]]],
           cliques[union[cart[fix[x_], V], complement[inverse[x_]]]]] :=
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]
```

Theorem. Since classes of cliques are hereditary, the same holds for the class of sets for which the symmetric restriction of **x** is reflexive.

```
In[23]:= SubstTest[image, inverse[S], intersection[cliques[u], cliques[v]],
           {u -> union[cart[fix[x], V], complement[x]],
            v -> union[cart[fix[x], V], complement[inverse[x]]]}] // Reverse

Out[23]= image[inverse[S], fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]] ==
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]

In[24]:= image[inverse[S], fix[image[inverse[CART], image[inverse[IMAGE[id[x_]]], RFX]]]] :=
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]
```

Lemma.

```
In[28]:= SubstTest[Uchains, cliques[t], t → intersection[x, y]] // Reverse

Out[28]= Uchains[intersection[cliques[x], cliques[y]]] == intersection[cliques[x], cliques[y]]

In[29]:= Uchains[intersection[cliques[x_], cliques[y_]]] := intersection[cliques[x], cliques[y]]
```

Theorem. Since classes of cliques are closed under chains of unions, the same holds for the class of sets for which the symmetric restriction of **x** is reflexive.

```
In[31]:= SubstTest[Uchains, intersection[cliques[u], cliques[v]],
           {u -> union[cart[fix[x], V], complement[x]],
            v -> union[cart[fix[x], V], complement[inverse[x]]]}] // Reverse

Out[31]= Uchains[fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]] ==
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]

In[32]:= Uchains[fix[image[inverse[CART], image[inverse[IMAGE[id[x_]]], RFX]]]] :=
          fix[image[inverse[CART], image[inverse[IMAGE[id[x]]], RFX]]]
```

Suppes' problem 1, part (e) on page 78.

On page 78, problem 1, part (e), the reader is asked to show that a symmetric, transitive relation is reflexive. This statement holds for the predicate **REFLEXIVE** as defined in the **GOEDEL** program.

```
In[33]:= implies[and[SYMMETRIC[x], TRANSITIVE[x]], REFLEXIVE[x]]

Out[33]= True
```

A version of this statement for restrictions of a relation **x** to a class **y** also holds, but a corresponding statement fails for the strong version of reflexivity, as shown by a counterexample.

Lemma. (Temporary rewrite rule.)

```
In[34]:= equal[composite[id[y], x, id[y]], composite[id[y], inverse[x], id[y]]] // AssertTest

Out[34]= equal[composite[id[y], x, id[y]], composite[id[y], inverse[x], id[y]]] ==
          subclass[composite[id[y], inverse[x], id[y]], x]
```

```
In[35]:= equal[composite[id[y_], x_, id[y_]], composite[id[y_], inverse[x_], id[y_]]] :=
          subclass[composite[id[y], inverse[x], id[y]], x]
```

Theorem. If the restriction of a relation is symmetric and transitive, then it is reflexive.

```
In[36]:= Map[implies[#, REFLEXIVE[composite[id[y], x, id[y]]]] &,
           SubstTest[and, SYMMETRIC[t], TRANSITIVE[t], t → restrict[x, y, y]]] // Reverse

Out[36]= or[not[subclass[composite[id[y], inverse[x], id[y]], x]],
           not[TRANSITIVE[composite[id[y], x, id[y]]]],
           REFLEXIVE[composite[id[y], x, id[y]]]] == True
```

```
In[37]:= or[not[subclass[composite[id[y_], inverse[x_], id[y_]], x_]],
           not[TRANSITIVE[composite[id[y_], x_, id[y_]]]],
           REFLEXIVE[composite[id[y_], x_, id[y_]]]] := True
```

Counterexample. If the restriction of a relation is symmetric and transitive, it need not be reflexive in the stronger sense.

```
In[38]:= (or[not[subclass[composite[id[y], inverse[x], id[y]], x]],
            not[TRANSITIVE[composite[id[y], x, id[y]]]],
            subclass[y, fix[x]]]) /. {x → EQUIDIFF, y → V}

Out[38]= False
```

Corollary. For the case of **y = udora[x]**, one finds:

```
In[39]:= SubstTest[or, not[subclass[composite[id[y], inverse[x], id[y]], x]],
           not[TRANSITIVE[composite[id[y], x, id[y]]]],
           REFLEXIVE[composite[id[y], x, id[y]]], y → udora[x]] // Reverse

Out[39]= or[not[subclass[inverse[x], x]],
           not[TRANSITIVE[composite[Id, x]]], REFLEXIVE[composite[Id, x]]] == True
```

```
In[40]:= or[not[subclass[inverse[x_], x_]],
           not[TRANSITIVE[composite[Id, x_]]], REFLEXIVE[composite[Id, x_]]] := True
```